
DUAL_DIFF_Thd_Temp  
*                                     ^  VC
*                            R1      /_\     R2
*                                     |
*                       __/\  /\  /\_ |__/\  /\  /\__
*                      |    \/  \/         \/  \/    |
*                  VC1 |                         VC2 |
*                      |______                _______| 
*                      |      |              |       |
*               VB1   _|     _|              |_      |_
*                ___|'_____|'                  `|______`|__
*     VIN      _|_  |`->   |`->NX          NX<-'|  1X<-'|  |
*      ______ /   \ QN1|  QN2 |           QN3|    QN4|     |
*     |VAC1  |EGin |   |      |_____________/|\______|    _|_
*    _|_    _|     |   |_____________________|   | VE2    ///
*   /AC1\   | \___/        _|_    VE1           _|_
*   \___/   |___|         / _ \I1              / _ \I2
*     |        _|_        \/ \/                \/ \/
*    _|_       ///        /\_/\                /\_/\
*    ///                  \___/                \___/
*             OUT          _|_                  _|_
*                          ///                  ///

*              _____/\  _____
*            _|_  R33 \/1K  _|_
*     VC1__ /   \           ///
*          |Eg1  |E_GAIN1
*     VC2__|     |
*           \___/       www.idea2ic.com dsauersanjose@aol.com  4/15/08
*            _|_
*            ///
*
.OPTIONS  GMIN=1e-18    METHOD=trap   srcsteps = 1  gminsteps = 1
*======== ====== ====== ====== ====== ====== ====== ====== ======
VCC       VC     0      DC     10
VAC1      VIN    0      DC     0      SIN(   0      1   1000 )
I1        VE1    0      1u
I2        VE2    0      1u
QN1       VC1    VB1    VE1    NPN1   1.00
QN2       VC1    VB1    VE2    NPN1   4.25
QN3       VC2    0      VE1    NPN1   4.25
QN4       VC2    0      VE2    NPN1   1.00
R1        VC1    VC     52K
R2        VC2    VC     52K
E_GAIN1   OUT    0      VC1    VC2    1
E_GAININ  VB1    0      VIN    0      1m

.control
tran             25u                     5m         0    1u
plot             out
echo            "THD% versus VIN_vpk and Temp_C"
setplot          new
set              NameList =  (       minus55C plus25C plus125C )
compose          TempVals    values  (-55)    25      125
compose          VinVals     values  10m 15m  20m 25m 30m 35m 40m 45m 50m 60m 70m 80m 90m
settype          voltage     VinVals
let              NoOfTemp =  length(TempVals)
let              NoOfVin =   length(VinVals)

begin
unset            interrupt
* ===============Loop_Temp============
let              j         =  1
while            (j       <=  NoOfTemp )
let              Tmp       =  TempVals[j-1]
set              temp      =  $&Tmp
set              thisName  =  $NameList[$&j]
let              $thisName =  0 * vector(NoOfVin)

* ===============Loop_Vin==============
let              k             = 1
while            (k           <= NoOfVin )
let              Vin           = VinVals[k-1]
alter            e_gainin gain = $&Vin
tran             25u             5m        0    1u
linearize
set              specwindow=    "blackman"
spec             200    8k       200     v(out)
let              thdsq =mag(out[9])^2 +mag(out[14])^2  +mag(out[19])^2  +mag(out[24])^2
let              thd_percent= 100*sqrt(thdsq)/mag(out[4]) 
echo             "$&unknown.Vin       $&thd_percent"
let              unknown.{$thisName}[unknown.k-1] = thd_percent
repeat           3



destroy
end
if               ($?interrupt)
goto             bail
endif
let              k =          k + 1
endwhile
setscale         VinVals
plot             $NameList loglog title "DUAL_DIFF THD_% vs Vin_pK and Temp_C"
let              j =          j + 1
endwhile
label            bail
echo            "Done."
end
.endc

*======== ====== ====== ====== ====== ====== ====== ====== ======
.model    NPN1   NPN(   BF=2100 VAF=216 )
.model    PNP1   PNP(   BF=2100  VAF=21)
.end

=====================END_OF_SPICE=======================

To Covert PDF to plain text click below
http://www.fileformat.info/convert/doc/pdf2txt.htm

This simulation only works on MacSpice for now.
Data in spice apparently gets stored in vectors which
are ready to be ploted. 

*                                     ^  VC
*                            R1      /_\     R2
*                                     |
*                       __/\  /\  /\_ |__/\  /\  /\__
*                      |    \/  \/         \/  \/    |
*                  VC1 |                         VC2 |
*                      |______                _______| 
*                      |      |              |       |
*               VB1   _|     _|              |_      |_
*                ___|'_____|'                  `|______`|__
*     VIN      _|_  |`->   |`->NX          NX<-'|  1X<-'|  |
*      ______ /   \ QN1|  QN2 |           QN3|    QN4|     |
*     |VAC1  |EGin |   |      |_____________/|\______|    _|_
*    _|_    _|     |   |_____________________|   | VE2    ///
*   /AC1\   | \___/        _|_    VE1           _|_
*   \___/   |___|         / _ \I1              / _ \I2
*     |        _|_        \/ \/                \/ \/
*    _|_       ///        /\_/\                /\_/\
*    ///                  \___/                \___/
*             OUT          _|_                  _|_
*                          ///                  ///

The distortion method does a transient analysis at various 
input levels and temperatures. The distortion is found by 
doing an RMS sum of the harmonics and then dividing by the 
fundamental.



Plotting distortion versus input level shows that
the distortion is scaled to absolute temperature.


